
# (* .................................... BASICS *) ;;
# () ;;
- : unit = ()
# 3 ;;
- : int = 3
# 3.0 ;;
- : float = 3.
# ’A’ ;;
- : char = ’A’
# "xyz" ;;
- : string = "xyz"
# false ;;
- : bool = false
# Some 3 ;;
- : int option = Some 3
# None ;;
- : ’a option = None
# [3; 4] ;;
- : int list = [3; 4]
# [] ;;
- : ’a list = []
# (2, "xyz", 3.0) ;;
- : int * string * float = (2, "xyz", 3.)
# (* .................... CHARACTERS AND STRINGS *) ;;
# Char.code ’a’ ;;
- : int = 97
# Char.code ’A’ ;;
- : int = 65
# Char.code ’0’ ;;
- : int = 48
# Char.chr 97 ;;
- : char = ’a’
# String.length "hello" ;;
- : int = 5
# (* .................................PERVASIVES *) ;;
# abs (-42) ;;
- : int = 42
# 42 mod 8 ;;
- : int = 2
# float_of_int 42 ;;
- : float = 42.
# 2. ** 4. ;;
- : float = 16.
# ceil 42.8 ;;
- : float = 43.
# floor 42.8 ;;
- : float = 42.
# log 42. ;;
- : float = 3.73766961828336841
# sqrt 42. ;;
- : float = 6.48074069840786
# int_of_float 17.7 ;;
- : int = 17
# 3 < 5 && true ;;

- : bool = true
# min 3 5, max 3 5 ;;
- : int * int = (3, 5)
# fst (2,"abc") ;;
- : int = 2
# snd (2, "abc") ;;
- : string = "abc"
# "x" ^ "y" ^ "z" ;;
- : string = "xyz"
# let x = 3 in
if x < 0 || x > 0 then "nonzero"
else "zero" ;;
- : string = "nonzero"
# let e = exp 1. in
let pi = 2. *. asin 1. in
Printf.printf "e = %f7.6; pi = %f7.6\n" e pi ;;
e = 2.7182827.6; pi = 3.1415937.6
- : unit = ()
# compare 2 4, compare 4 2, compare 3 3 ;;
- : int * int * int = (-1, 1, 0)
# (* ................. FUNCTIONS AND APPLICATION *) ;;
# fun x -> x + 1 ;;
- : int -> int = <fun>
# (fun x -> x + 1) 3 ;;
- : int = 4
# fun x y -> x + y ;;
- : int -> int -> int = <fun>
# fun (x, y) -> x + y ;;
- : int * int -> int = <fun>
# fun () -> 4 ;;
- : unit -> int = <fun>
# let uncurriedplus (x, y) = x + y in
let curriedplus x y = x + y in
(uncurriedplus (1, 2), curriedplus 1 2) ;;

- : int * int = (3, 3)
# (* .................................. MATCHING *) ;;
# let x = 3 in
match x with
| 0 -> "zero"
| 1 -> "one"
| _ -> "more than one" ;;

- : string = "more than one"
# let x, y = Some 111, 2999 in
match (x, y) with
| Some z, _ -> z + y
| None, _ -> y ;;

- : int = 3110
# let rec sum (x : int list) : int =
match x with
| [] -> 0
| u :: t -> u + sum t ;;

val sum : int list -> int = <fun>
# (* ................................... RECORDS *) ;;
# type rcrd = {foo : int; bar : string} ;;

type rcrd = { foo : int; bar : string; }
# {foo = 3; bar = "xyz"} ;;
- : rcrd = {foo = 3; bar = "xyz"}
# (* ..................................... LISTS *) ;;
# 3 :: [4; 5] ;;
- : int list = [3; 4; 5]
# [1;2;3] @ [4;5;6] ;;
- : int list = [1; 2; 3; 4; 5; 6]
# List.length [8; 9; 10] ;;
- : int = 3
# List.hd [3; 4] ;;
- : int = 3
# List.tl [3 ;4] ;;
- : int list = [4]
# List.tl [4] ;;
- : int list = []
# List.map ;;
- : (’a -> ’b) -> ’a list -> ’b list = <fun>
# List.map (fun x -> x + 100) [2;3;4] ;;
- : int list = [102; 103; 104]
# List.map (fun x -> x = 3) [2;3;4] ;;
- : bool list = [false; true; false]
# List.filter ;;
- : (’a -> bool) -> ’a list -> ’a list = <fun>
# List.filter (fun x -> x < 4) [4;3;9;6;1;0;5] ;;
- : int list = [3; 1; 0]
# List.fold_right ;;
- : (’a -> ’b -> ’b) -> ’a list -> ’b -> ’b = <fun>
# List.fold_right (^) ["a";"b";"c"] "x" ;;
- : string = "abcx"
# List.fold_left ;;
- : (’a -> ’b -> ’a) -> ’a -> ’b list -> ’a = <fun>
# List.fold_left (^) "x" ["a";"b";"c"] ;;
- : string = "xabc"
# List.find ;;
- : (’a -> bool) -> ’a list -> ’a = <fun>
# List.find (fun x -> x > 10) [1;5;10;13;19] ;;
- : int = 13
# List.find (fun x -> x > 10) [1;5;10] ;;
Exception: Not_found.
# List.rev [8; 9; 10] ;;
- : int list = [10; 9; 8]
# List.nth [8; 9; 10] 2 ;;
- : int = 10
# List.nth [8; 9; 10] 3 ;;
Exception: Failure "nth".
# List.sort compare [5; 2; 4] ;;
- : int list = [2; 4; 5]
# (* ........................ VARIANT DATA TYPES *) ;;
# type ’a option =
| Some of ’a
| None ;;

type ’a option = Some of ’a | None
# type ’a stack =
| Empty
| Top of (’a * ’a stack) ;;

type ’a stack = Empty | Top of (’a * ’a stack)
# Top (3, Empty) ;;
- : int stack = Top (3, Empty)
# (* ................................ EXCEPTIONS *) ;;
# Not_found ;;
- : exn = Not_found
# raise Not_found ;;
Exception: Not_found.
# raise (Failure "error") ;;

Exception: Failure "error".
# failwith "error" ;;
Exception: Failure "error".
# try
Some (List.find (fun x -> x > 10) [1;5;10])

with
| Not_found -> None
| e -> raise e ;;

- : int option = None
# (* .................... MODULES AND SIGNATURES *) ;;
# module type STACK =
sig
type elt
type stack
exception Empty of string
val empty : unit -> stack
val push : stack * elt -> stack
val pop : stack -> elt * stack
val isEmpty : stack -> bool

end ;;
module type STACK =
sig
type elt
type stack
exception Empty of string
val empty : unit -> stack
val push : stack * elt -> stack
val pop : stack -> elt * stack
val isEmpty : stack -> bool

end
# module MakeStack(Arg: sig type t end)

: (STACK with type elt = Arg.t) =
struct
type elt = Arg.t
type stack = elt list
exception Empty of string
let empty () = []
let push (s,x) = x::s
let pop s = match s with
| x :: t -> (x, t)
| [] -> raise (Empty "empty")
let isEmpty = fun x -> x = []

end ;;
module MakeStack :
functor (Arg : sig type t end) ->
sig
type elt = Arg.t
type stack
exception Empty of string
val empty : unit -> stack
val push : stack * elt -> stack
val pop : stack -> elt * stack
val isEmpty : stack -> bool

end
# module IntStack = MakeStack(struct type t = int end) ;;
module IntStack :
sig
type elt = int
type stack
exception Empty of string
val empty : unit -> stack
val push : stack * elt -> stack
val pop : stack -> elt * stack
val isEmpty : stack -> bool

end

# IntStack.push (IntStack.empty () , 3) ;;
- : IntStack.stack = <abstr>
# (* ......................... REFS AND MUTABLES *) ;;
# let rint = ref 3 ;;
val rint : int ref = {contents = 3}
# rint := !rint + 5 ;;
- : unit = ()
# rint ;;
- : int ref = {contents = 8}
# !rint ;;
- : int = 8
# type mut_demo = { mutable mut: int ; nonmut: int } ;;
type mut_demo = { mutable mut : int; nonmut : int; }
# let m = { mut = 3; nonmut = 4 } ;;
val m : mut_demo = {mut = 3; nonmut = 4}
# m.mut <- 5 ;;
- : unit = ()
# m ;;
- : mut_demo = {mut = 5; nonmut = 4}
# (* .................................. LAZINESS *) ;;
# let a = lazy (!rint * 2) ;;
val a : int lazy_t = <lazy>
# rint := 42 ;;
- : unit = ()
# Lazy.force a ;;
- : int = 84
# (* ....................... OBJECTS AND CLASSES *) ;;
# class type demo_parent =
object
val v1 : int
val mutable v2 : bool
method getv1 : int
method getv2 : bool
method setv2 : bool -> unit

end ;;
class type demo_parent =
object
val v1 : int
val mutable v2 : bool
method getv1 : int
method getv2 : bool
method setv2 : bool -> unit

end
# class type demo_child =
object
inherit demo_parent
method getv3 : float (* no exposed inst var *)

method setv3 : float -> unit
end ;;

class type demo_child =
object
val v1 : int
val mutable v2 : bool
method getv1 : int
method getv2 : bool
method getv3 : float
method setv2 : bool -> unit
method setv3 : float -> unit

end
# class demo : demo_child =
object
val v1 = 0
val mutable v2 = true
val mutable v3 = 0.0
method getv1 = v1
method getv2 = v2
method setv2 b = v2 <- b
method getv3 = v3
method setv3 f = v3 <- f

end ;;
class demo : demo_child
# let ob = new demo ;;
val ob : demo = <obj>
# ob#getv1, ob#getv2, ob#getv3 ;;
- : int * bool * float = (0, true, 0.)
# ob#setv2 false; ob#setv3 3.14 ;;
- : unit = ()
# ob#getv1, ob#getv2, ob#getv3 ;;
- : int * bool * float = (0, false, 3.14)
# (* .................................. PRINTING *) ;;
# print_int 42; print_newline () ;;
42
- : unit = ()
# print_float 3.14159; print_newline () ;;
3.14159
- : unit = ()
# print_string "a string\n" ;;
a string
- : unit = ()
# Printf.printf "int: %d; float: %f; bool: %B; string: %s\n"
42 3.14159 true "a string" ;;

int: 42; float: 3.141590; bool: true; string: a string
- : unit = ()


